Ecological Archives - A1

Nice et al. 2013. A Hierarchical Perspective on the Diversity of Butterfly Species' Responses to Weather in the Sierra Nevada Mountains

Appendix A. Example of BUGS code.

An example of model specification for BUGS used for hierarchical analyses of individual climate covariates (plus a year effect) is provided below. This code was used for the analysis of the Donner Pass data with covariates spring minimum temperature and year. Otherwise identical code was used for all analyses. The first model block is the code for the unconstrained model in which species are allowed to have different β coefficients. This is followed by the constrained model in which species are constrained to have identical β coefficients. The multivariate analyses used an expanded version of this code.

Unconstrained model:

```
model{
# binomial likelihood for occurrence
# and logit link function for glm
for(i in 1:N){
      DPs[i] ~ dbin(p[i], Visits[i])
# inverse logit
      p[i] <- 1 / (1 + exp(-1 * alpha[i]))
      alpha[i] <- mu[Sp[i]] + beta1[Sp[i]] * Std_Sp_minT[i] + beta2[Sp[i]] * Std_Year[i]
}
# random effect (hierarchical) coefficients for individual species (conditional priors)
for(j in 1:Nsp){
     beta1[j] ~ dnorm(beta1mu, beta1tau)
     beta2[j] ~ dnorm(beta2mu, beta2tau)
     mu[j] ~ dnorm(mumu, mutau)
}
# uninformative precision hyperpriors
beta1tau ~ dgamma(0.1,0.001)
beta2tau ~ dgamma(0.1,0.001)
```

```
mutau ~ dgamma(0.1,0.001)
# uninformative mean hyperpriors
beta1mu ~ dnorm(0,0.00001)
beta2mu ~ dnorm(0,0.00001)
mumu ~ dnorm(0,0.00001)
}
```

Constrained model:

```
model{
# binomial likelihood for occurrence
# and logit link function for glm
for(i in 1:N){
      DPs[i] ~ dbin(p[i], Visits[i])
# inverse logit
      p[i] <- 1 / (1 + exp(-1 * alpha[i]))
      alpha[i] <- mu + beta1 * Std_Sp_minT[i] + beta2 * Std_Year[i]
}
# uninformative priors for
beta1 ~ dnorm(0,0.001)
beta2 ~ dnorm(0,0.001)
mu ~ dnorm(0,0.00001)
}
```